
Crésus Salaires

Crésus Salaires

© 2025 - EPSITEC1/18



19.1 - L'éditeur de calcul
19.1.1 - Quelques principes de base

19.1.2 - Les fonctions "Opération"

19.1.3 - Les fonctions "Test"

19.1.4 - Les fonctions "Conversion"

19.1.5 - Les fonctions "Pourcentage"

19.1.6 - Les fonctions "Date"

19.1.7 - Les fonctions "Chaîne"

19.1.8 - Concaténer

Crésus Salaires

© 2025 - EPSITEC2/18



19.1 - L'éditeur de calcul

L’éditeur de calcul de Crésus Salaires permet de créer des formules personnalisées 

pour obtenir des résultats précis : calculer un montant, afficher un texte 

conditionnel, insérer une date, ou combiner diverses informations provenant de 

différentes rubriques.

L’éditeur de calcul affiche trois sections :

La zone de formule (en haut) : champ dans lequel on écrit le calcul. On 

combine des chiffres, des opérations, des fonctions (par ex. SOMME() ou 

SI()), ou encore des rubriques prédéfinies (comme Salaire mensuel ou 

Nom de l’employé).

Opérations (au centre) : liste de fonctions disponibles, classées par type.

Variables (en bas) : ensemble des rubriques disponibles dans Crésus. 

Double-cliquer sur une rubrique ou une fonction l’insère directement dans 

la zone de formule.

Ce chapitre vous présente une grande partie des fonctions générales de l’éditeur de 

calcul. D’autres fonctions moins courantes sont également décrites dans 

cet article de notre base de connaissance.

19.1.1 - Quelques principes de base

Les nombres s’écrivent manuellement et les décimales sont séparées par 

un point (pas une virgule) : 2000 * 0.08 calcule 8 % de 2000.

Les chaînes de texte (mots ou phrases) doivent être saisies entre 

guillemets : "Bonjour" ou "à verser".

Les rubriques doivent être insérées telles qu’elles apparaissent dans la 

liste (par ex. @Ind:Salaire mensuel), nous vous déconseillons donc de 

les saisir manuellement.

Les espaces ne sont pas pris en compte dans le calcul mais rendent la 

lecture plus claire (2000+300 = 2000 + 300). Si vous souhaitez inclure 

un espace dans le résultat du calcul (par exemple entre le prénom et le 

nom de l’employé), il faut saisir dans la zone de formule un espace entre 

guillemet : " ".

Les sauts de ligne ne changent pas le calcul, mais permettent d’aérer les 

longues formules. Si vous souhaitez inclure un saut de ligne dans le 

résultat du calcul (par exemple entre un nom et une adresse), il faut saisir 

dans la zone de formule la fonction FINLIGNE() ou "\n".

19.1.2 - Les fonctions "Opération"

Les opérateurs arithmétiques :

Dans l’éditeur de calcul, les opérateurs suivants servent aux calculs numériques :

+ additionne deux valeurs. Exemple : le salaire de base + une prime.

– soustrait une valeur. Exemple : le salaire brut – une retenue.

* multiplie deux valeurs. Exemple : nombre d’heures * taux horaire.

/ divise une valeur par une autre. Exemple : montant / nombre de jours.

Les fonctions mathématiques et statistiques :

Ces fonctions permettent d’affiner ou de contrôler les calculs numériques. Elles sont 

utiles par exemple pour arrondir un résultat, comparer plusieurs montants ou 

manipuler des valeurs intermédiaires.

MIN(valeur1 ; valeur2) : retourne la plus petite des deux valeurs.

Exemple : MIN(1000 ; 1200) retourne « 1200 ».

MAX(valeur1 ; valeur2) : retourne la plus grande des deux valeurs.

Exemple : MAX(1000 ; 1200) retourne « 1200 ».

SOMME(valeur1 ; valeur 2 ; ...) : retourne la somme d’une liste 

de nombres.

Exemple : SOMME(12;45;-3;28) retourne « 82 ».

MOYENNE(valeur1 ; valeur2 ; …) : calcule la moyenne des valeurs 

indiquées.

Exemple : MOYENNE(100 ; 200 ; 300) retourne « 200 ».

Les totalisateurs :

Les fonctions de totalisation permettent de cumuler des valeurs sur plusieurs 

dimensions : tous les employés, une période donnée (année, trimestre, mois 

précédent), ou encore selon des critères particuliers. Elles ne renvoient donc pas 

uniquement la valeur d’une rubrique pour un employé et un mois donnés, mais 

calculent des sommes globales.

TOTEMPLOYES()

Additionne les valeurs d’une rubrique pour tous les employés.

Exemple : obtenir le total des salaires bruts versés dans l’entreprise.

TOTANNEE()

Calcule le total depuis le début de l’année jusqu’au mois courant, en tenant 

compte uniquement des périodes où un salaire a été calculé.

Exemple : total des cotisations AC pour l’année en cours.

TOUTANNEE()

Comme TOTANNEE, mais inclut également les mois sans salaire. On peut 

ainsi ajouter une valeur par défaut aux périodes vides.

TOTTRIMESTRE()

Cumule les valeurs sur le trimestre où se situe le mois sélectionné.

Exemple : total des remboursements kilométriques pour le trimestre.

PRECEDENT()

Donne la valeur d’une rubrique pour la période précédente (y compris celle 

de l’année passée si l’on est en janvier).

Exemple : montant du salaire versé le mois dernier.

TOTPRECEDENT()

Additionne toutes les valeurs des périodes écoulées.

La fonction CUMUL() :

Cette fonction additionne toutes les indemnités selon leur assujettissement à un 

brut déterminant ou selon leur affectation dans le certificat de salaire.

Par exemple :

CUMUL("AVS") donne le total soumis à l’AVS.

CUMUL("2.1") donne le total des montants reportés en case 2.1 du 

certificat de salaire.

Les critères utilisables sont nombreux (BRUT, NET, VERSE, AVS, AC, LAA, LPP, 

etc.).

On peut combiner plusieurs critères avec ET (&) ou OU (|), ou exclure un critère 

avec ! (par ex. !AVS pour les indemnités non soumises à l’AVS).

19.1.3 - Les fonctions "Test"

La fonction SI() :

La fonction SI() permet de définir un résultat différent selon qu’une condition est 

vraie ou fausse. Elle s’écrit de la manière suivante :

SI(condition ; valeur si vrai ; valeur si faux)

condition : une comparaison ou une combinaison de comparaisons (par 

ex. Salaire > 0).

valeur si vrai : le résultat à afficher si la condition est vraie.

valeur si faux : le résultat à afficher si la condition est fausse.

Exemple 1 (texte selon salaire) :

SI(@Ind:Salaire mensuel > 0 ; "à verser" ; "à retenir")

Résultat : affiche « à verser » si le salaire est positif, sinon « à retenir ».

Exemple 2 (montant minimum) :

SI(@Ind:Salaire mensuel < 2000 ; 2000 ; @Ind:Salaire mensuel)

Résultat : si le salaire est inférieur à 2000, retourne 2000, sinon le salaire réel.

Exemple 3 (condition combinée) :

SI((@Ind:Salaire mensuel > 0) ET (@Ind:Taux AVS <> 0) ; 

"cotisation due" ; "aucune cotisation")

Résultat : affiche « cotisation due » si le salaire est positif et que le taux AVS est 

différent de zéro.

Les opérateurs de comparaison :

Ces opérateurs permettent de tester une condition :

= signifie « égal à ». Exemple : vérifier si une rubrique est égale à 0.

<> signifie « différent de ».

< signifie « plus petit que ». Exemple : comparer un taux avec un seuil 

minimal.

<= signifie « plus petit ou égal à ».

> signifie « plus grand que ». Exemple : tester si le nombre d’heures est 

supérieur à 40.

>= signifie « plus grand ou égal à ».

Les opérateurs logiques :

Une comparaison retourne toujours une valeur logique : vrai ou faux.

Il est possible d’associer plusieurs conditions grâce aux opérateurs logiques :

ET vérifie que deux conditions sont vraies en même temps.

OU vérifie qu’au moins une des conditions est vraie.

NON inverse le résultat d’une condition (vrai devient faux, et inversement).

Par exemple :

(@Ind:Salaire mensuel > 0) ET (@Ind:Taux AVS <> 0)

Résultat : cette condition retourne VRAI si le salaire est positif et si le taux AVS est 

différent de zéro.

19.1.4 - Les fonctions "Conversion"

ARRONDI(valeur ; décimales) : arrondit une valeur au nombre de 

décimales souhaité.

Exemple : ARRONDI(123.456 ; 2) retourne « 123.46 ».

INT(valeur) : retourne uniquement la partie entière d’un nombre.

Exemple : INT(123.456) retourne « 123 ».

FRAC(valeur) : retourne uniquement les décimales d’un nombre.

Exemple : FRAC(12.775) retourne « 0.775 ».

Les fonctions DEFAUT() et VIDE() :

DEFAUT(rubrique ; valeur de substitution) : si la rubrique 

spécifiée est vide, la remplace par la valeur de substitution indiquée.

Exemple : DEFAUT(@Ind:Salaire mensuel ; 100) retourne le montant du 

salaire s’il y en a un, sinon « 100 ».

VIDE() : retourne un blanc (vide) au lieu d’une valeur.

Exemple : SI(@Coef:Jours travaillés>0 ; @Coef:Jours travaillés 

; VIDE()) n’affiche rien plutôt que « 0 ».

SI(VIDE(rubrique);valeur si vrai; valeur si faux) : 

retourne la valeur 1 (vrai) si la rubrique est vide, sinon la valeur 0 (faux).

Exemple : SI(VIDE(@Ind:Salaire mensuel); "à l’heure"; "au mois")

retourne « à l’heure » si le salaire mensuel est vide, « au mois » dans le cas 

contraire.

Les fonctions CHAINE() :

Pour insérer une rubrique dans un texte, il faut distinguer deux cas :

Si la rubrique est déjà du texte (nom, prénom, adresse), il n’est pas nécessaire 

de la convertir :

"Employé : " + @Donnée:Nom de l'employé

Résultat : « Employé : Dupont ».

Si la rubrique est un nombre ou une date (salaire, taux, date d’entrée), il faut 

convertir la valeur en texte (ou chaîne de caractères) avec la fonction CHAINE(). 

Cette fonction s’écrit de cette façon :

CHAINE(valeur ; décimales)

valeur : ce qui doit être converti en texte (par ex. un salaire, une date, un 

pourcentage).

décimales : le nombre de chiffres à afficher après la virgule.

Par exemple :

"Montant : " + CHAINE(@Ind:Salaire mensuel ; 2)

Résultat : « Montant : 4500.00 »

CHAINEMOIS(mois ; langue)

Retourne le nom du mois (de 1 à 12) et, éventuellement (le second paramètre est 

optionnel), dans la langue indiquée (0 = français, 1 = allemand, 2 = anglais, 3 = 

italien). Par défaut, la fonction donne le nom dans la langue du document.

Par exemple :

CHAINEMOIS(2)

Résultat : « février »

CHAINEMOIS(7 ; 1)

Résultat : « Juli »

CHAINEJDLS(jour ; langue)

Retourne le nom du jour de la semaine (de 1 à 7) et, éventuellement (le second 

paramètre est optionnel), dans la langue indiquée (0 = français, 1 = allemand, 2 = 

anglais, 3 = italien). Par défaut, la fonction donne le nom dans la langue du 

document.

Par exemple :

CHAINEJDLS(2)

Résultat : « mardi »

CHAINEJDLS(1;2)

Résultat : « Monday »

19.1.5 - Les fonctions "Pourcentage"

C%(montant ; taux) : Calcule le pourcentage d’un montant.

Exemple : C%(20 ; 7) retourne « 1.4 » (= 7% de 20)

A%(montant ; taux) : Ajoute un pourcentage à un montant.

Exemples : A%(20 ; 7) retourne « 21.4 » ; A%(20 ; -7) retourne « 18.6 »

S%(montant ; taux) : Calcule la valeur qu’avait un montant avant qu’on ne lui 

ajoute un pourcentage.

Exemple : S%(21.4 ; 7) retourne « 20 »

M%(montant ; taux) : Calcule la valeur ajoutée à un montant augmenté d’un 

pourcentage.

Exemple : M%(21.4 ; 7) retourne « 1.4 »

T%(montant 1 ; montant 2) : Calcule le pourcentage entre deux montants.

Exemple : T%(21.4 ; 20) retourne « 7 »

19.1.6 - Les fonctions "Date"

Ces fonctions permettent de travailler avec des dates : en extraire une partie (jour, 

mois, année), calculer une durée ou mettre une date en forme.

AUJOURDHUI() : retourne la date du jour (date système).

DATE(jour ; mois ; année) : retourne une date spécifiée au format 

jj.mm.aaaa.

Exemple : DATE(25;12;18) retourne « 25.12.2018 ».

JOUR(date) : retourne le jour du mois (1 à 31).

Exemple : JOUR(@Donnée:Date d’entrée) retourne « 15 » si la date est le 15.

MOIS(date) : retourne le mois (1 à 12).

Exemple : MOIS(@Donnée:Date d’entrée) retourne « 3 » si la date est en mars.

ANNEE(date) : retourne l’année sur 4 chiffres.

Exemple : ANNEE(@Donnée:Date d’entrée) retourne « 2025 ».

JDSL(date) : retourne le jour de la semaine (1 = lundi à 7 = dimanche) de 

la date indiquée.

Exemple : JDSL(DATE(28;3;14)) retourne « 5 » (vendredi).

SEMAINE(date) : convertir une date en numéro de semaine (0 à 53). Les 

décimales correspondent au nombre de jour en centièmes depuis le début 

de la semaine (lundi = 0 ; mardi = 0.14 ; etc. jusqu’à dimanche = 0.86). 

Vous pouvez bien sûr utiliser la fonction INT pour supprimer les décimales.

Exemple : SEMAINE(DATE(28;3;14)) retourne « 13.57 » (vendredi de la 13ème

semaine).

ADDMOIS(date ; nombre) : ajoute un ou plusieurs mois à une date 

donnée.

Exemple : ADDMOIS(DATE(15;5;25);2) retourne « 15.04.2025 ».

DEBUTMOIS(date) : retourne la date du premier jour du mois.

Exemple : DEBUTMOIS(DATE(3;2;25)) retourne « 01.02.2025 ».

FINMOIS(date) : retourne la date du dernier jour du mois.

19.1.7 - Les fonctions "Chaîne"

Ces fonctions permettent de manipuler du texte : transformer une chaîne de 

caractères, en extraire une partie ou la mettre en forme.

MAJUSCULE(texte) : affiche un texte en majuscules.

Exemple : MAJUSCULE("Dupont") retourne DUPONT.

MINUSCULE(texte) : affiche un texte en minuscules.

Exemple : MINUSCULE("Dupont") retourne dupont.

NOMPROPRE(texte) : Met la première lettre de chaque mot de la chaîne 

de caractères en lettre majuscule et les autres lettres en minuscules.

GAUCHE(texte ; nombre) : extrait le nombre de caractères souhaités 

en partant depuis la gauche.

Exemple : GAUCHE("Bonjour";3) retourne « Bon ».

DROITE(texte ; nombre) : extrait le nombre de caractères souhaités 

en partant depuis la droite.

MILIEU(texte ; position ; nombre) : extrait le nombre de 

caractères souhaités depuis la position indiquée (première lettre du texte = 

0).

LONGUEUR(texte): affiche le nombre de caractères (espaces compris) 

d’un texte.

Exemple : LONGUEUR("Jean") retourne 4.

TROUVE(texte ; critère) : indique à quelle position (en partant de 0) 

le critère souhaité se situe dans le texte (ou chaîne de caractères.

Exemple : TROUVE("Bonjour";"jou") retourne « 3 ».

REMPLACE(texte ; critère ; texte de substitution) : 

remplace le paramètre critère par un texte de substitution.

19.1.8 - Concaténer

Pour coller plusieurs textes ensemble, il suffit d’utiliser le signe + :

Exemple :

"Employé : " + @Donnée:Prénom + " " + MAJUSCULE(@Donnée:Nom)

"Employé : " = texte fixe.

@Donnée:Prénom = la rubrique contenant le prénom (ex. Jean).

" " = un espace inséré manuellement pour séparer prénom et nom.

@Donnée:Nom = la rubrique contenant le nom (ex. Dupond).

MAJUSCULE(@Donnée:Nom) = DUPOND

Résultat : « Employé : Jean DUPOND »

